

Code Review 4: Modular Programming

1 Modules

Definition 1.1 (Module). A **module** is the packaging of pieces of software such as functions and types.

A module is specified by placing the definitions of its keywords between the keywords `struct` and `end`:

```
module <modulename> =  
  struct  
    <definition1>  
    <definition2>  
    <definition3>  
    ...  
  end
```

We can define types for modules akin to how we define types for values and functions. A **module signature** defines the **interface** of a module, the types and values a module must provide.

To define a module signature, we use the following syntax, placing **type definitions** between the `sig` and `end` keyword:

```
module type <typename> =  
  sig  
    <defintion1>  
    <definition2>  
    <definition3>  
    ...  
  end
```

We can write that a module `<modulename>` follows a module signature `<moduletype>` by using the `:` operator, just as we do with OCaml expressions.

```
module <modulename> : <moduletype> =  
  struct  
    <definition1>  
    <definition2>  
    <definition3>  
    ...  
  end
```

Example 1.2 (Math Module). In lab 7, we defined a `Math` module that satisfies the `MATH` type signature.

Listing 1: Math Module

```

1 module type MATH =
2   sig
3     (* the constant pi *)
4     val pi : float
5     (* cosine of an angle in radians *)
6     val cos : float -> float
7     (* sine of an angle in radians *)
8     val sin : float -> float
9     (* sum of two numbers *)
10    val sum : float -> float -> float
11    (* maximum value in a list; None if list is empty *)
12    val max_opt : float list -> float option
13  end ;;
14
15
16 module Math : MATH =
17   struct
18     let pi = 3.14159
19     let cos = cos
20     let sin = sin
21     let sum = (+.)
22     let max_opt lst =
23       match lst with
24       | [] -> None
25       | hd :: tl -> Some (List.fold_left max hd tl)
26   end ;;

```

The `val` keyword is used to declare the types of values within a module signature.

Remark 1.3 (Opening a module). There are two ways that we'll typically open or use a module:

1. **Dot Notation.** For instance, we can define the we can use the `Math` module above to find the maximum element in a list.

```
Math.max_opt [5;9;3;2;1;6]
```

2. **Open keyword.** We can use the `open` keyword, and then use values from the module without prefixing with a dot notation and the module name.

```

(* Method 1 *)
let open Math in max_opt [5;9;3;2;1;6] ;;

(* Method 2 *)
open Math ;;
max_opt [5;9;3;2;1;6]

```

As a design and style note, we'll typically use `open` when we want to use the same values or functions from a module multiple times in a single file. However, even though prefixing with the

module name can be longer syntactically, when we may have conflicts between function or value names, it is better to use dot notation, which follows from the **edict of intention**.

2 Abstract Data Types

Definition 2.1 (Abstract Data Type). An **Abstract Data Type** (ADTs) is a type for objects whose behavior is **defined by its values and operations**. The definition of an ADT only provides the **type signatures of values and operations**, **not** how these operations are **implemented**. They are called "abstract" because the implementation details are hidden. ¹

Modules allow us to implement abstract data types and enforce **invariants**, properties of the ADTs that must hold true.

Example 2.2 (Stack). An example of an abstract data type is a **stack**, a collection of elements that enforces the **Last-in-First-Out (LIFO)** invariant, which means that the last element *pushed* on to the stack must be the first element *popped* from the stack.

We can define a stack to support the following values and operations:

- `empty` : A stack with no elements
- `PUSH(i, stack)` : Pushes an element `i` onto `stack`
- `POP(stack)` : Pops an element from a `stack`
- `TOP(stack)` : Returns the element at the top of the `stack`

In lab 7, we first tried implementing a stack of integers, **internally as a list**.

Listing 2: IntListStack

```
1 module IntListStack =
2   struct
3     exception EmptyStack
4
5     type stack = int list
6
7     let empty : stack = []
8
9     let push (i : int) (s : stack) : stack =
10       i :: s
11
12     let top (s : stack) : int =
13       match s with
14       | [] -> raise EmptyStack
15       | h :: _ -> h
16
17     let pop (s : stack) : stack =
18       match s with
19       | [] -> raise EmptyStack
20       | _ :: t -> t
21   end ;;
```

¹<https://www.geeksforgeeks.org/abstract-data-types/>

There's a problem with this implementation. The **internal implementation of a stack is not hidden to a user of the `IntListStack` module**. Therefore, a user can break the LIFO invariant by calling a function like `List.rev`.

As shown in lab 7, the following code would break the invariant:

Listing 3: Bad Stack

```
1
2 let small_stack () : IntListStack.stack =
3   let open IntListStack in
4     empty
5     |> push 5
6     |> push 1 ;;
7
8 let invert_stack : IntListStack.stack -> IntListStack.stack =
9   List.rev ;;
10
11 let bad_el = IntListStack.top (invert_stack (small_stack ())) ;;
```

Here, defining `bad_el` breaks the LIFO invariant. `invert_stack` **directly** reverses the elements in the stack; we shouldn't be able to invert a stack without pushing and popping the elements in last-in-first-out order (see **Problem 2.3**).

Problem 2.3. Create a proper implementation of a stack that allows you to correctly reverse the elements of a stack without breaking the LIFO invariant.

We can avoid this problem by **using a module signature**, proving an **abstract data type** that a user of the module can use.

Listing 4: INT_STACK

```
1 module type INT_STACK =
2   sig
3     exception EmptyStack
4     type stack
5     val empty : stack
6     val push : int -> stack -> stack
7     val top : stack -> int
8     val pop : stack -> stack
9   end ;;
```

Here, we **hide the internal implementation** of the `stack` abstract data type behind an **abstraction barrier** enforced using the `INT_STACK` module type. In other words, we provide the **abstract** data type `stack` without revealing the **concrete** implementation of the data type.

Using this module type signature, we can redefine or `IntListStack` module.

Listing 5: SafeIntListStack

```

1 module SafeIntListStack = (IntListStack : INT_STACK) ;;
2
3 let example_stack : IntListStack.stack =
4   let open IntListStack in
5     empty
6     |> push 5
7     |> push 1 ;;
8
9 // This would result in an error! Try testing this in the REPL!!
10 let example_element : int =
11   example_stack
12   |> List.rev
13   |> IntListStack.top ;;

```

3 Polymorphic Modules

We can also define modules that implement **polymorphic** abstract data types.

Here's how we would define stacks, by providing the polymorphic abstract data type `'a stack`:

Listing 6: Polymorphic Stack

```

1 module type STACK =
2   sig
3     exception EmptyStack
4     type 'a stack
5     val empty : 'a stack
6     val push : 'a -> 'a stack -> 'a stack
7     val top : 'a stack -> 'a
8     val pop : 'a stack -> 'a stack
9   end ;;
10
11
12 module Stack : STACK =
13   struct
14     exception EmptyStack
15
16     type 'a stack = 'a list (* We've chosen to implement stacks
17                               internally as lists, a natural
18                               and simple choice *)
19
20     let empty : 'a stack = []
21
22     (* push i s -- Adds an element i to the top of stack s *)
23
24     let push (elt : 'a) (stk : 'a stack) : 'a stack =
25       elt :: stk

```

```

25
26 (* pop_helper s -- Returns a pair of the top element of the
27    stack and a stack containing the remaining elements *)
28 let pop_helper (stk : 'a stack) : 'a * 'a stack =
29     match stk with
30     | [] -> raise EmptyStack
31     | hd :: tl -> (hd, tl)
32
33 (* top s -- Returns the value of the topmost element on stack s,
34    raising the EmptyStack exception if there is no element to be
35    returned. *)
36 let top (stk: 'a stack) : 'a =
37     fst (pop_helper stk)
38
39 (* pop s -- Returns a stack with the topmost element from s
40    removed, raising the EmptyStack exception if there is no
41    element to be removed. *)
42 let pop (stk : 'a stack) : 'a stack =
43     snd (pop_helper stk)
44 end ;;

```

See Lab 7 for more details.

4 Sharing Constraints

Definition 4.1 (Sharing constraints). A **sharing constraint** adds additional information to a module signature, letting the user know about one or more type equalities. Consider Listing 4. Suppose, instead of a stack of `int`'s, we wanted to define a stack of `float`'s.

We could define a whole new signature for stacks of `float`'s.

Listing 7: `FLOAT_STACK`

```

1 module type FLOAT_STACK =
2   sig
3     exception EmptyStack
4     type stack
5     val empty : stack
6     val push : float -> stack -> stack
7     val top : stack -> float
8     val pop : stack -> stack
9   end ;;

```

However, this is repetitive. `INT_STACK` and `FLOAT_STACK` are identical, aside from type `int` being replaced with type `float`.

Instead, we can generalize this idea and create a `STACK` signature that works with any element type.

Listing 8: Stack

```

1 module type STACK =
2   sig
3     exception EmptyStack
4     type element
5     type stack
6     val empty : stack
7     val push : element -> stack -> stack
8     val top : stack -> element
9     val pop : stack -> stack
10  end ;;

```

Now, suppose we create a module that satisfies the STACK signature.

Listing 9: IntListStack

```

1 module IntListStack : STACK =
2   struct
3     exception EmptyStack
4     type element = int
5     type stack = int list
6     let empty : stack = []
7     let push elt s = elt :: s
8     let top s =
9       match s with
10      | [] -> raise EmptyStack
11      | h :: _ -> h
12     let pop s =
13       match s with
14      | [] -> raise EmptyStack
15      | _ :: t -> t
16   end ;;
17
18   let example_stack : IntListStack.stack =
19   let open IntListStack in
20     empty
21     |> push 5
22     |> push 1 ;;
23
24   (*
25
26   This gives us an error:
27
28   Error: This expression has type int but an expression was expected of
29   type IntListStack.element
30   *)

```

However, when we try to use the module, we get an error, even though the following is true:

The `STACK` signature says there will be a type `element` and a type `stack`, but does not say how these types are implemented. The `IntListStack` says that the type `element` will be an `int` and that the type `stack` will be an `element list`.

However, recall that users of our module only have access to what's provided in the type signature. So, the fact that `element` is an `int` is hidden behind the **abstraction barrier**, as the `element` is only defined to be an `int` in the module implementation in Listing 9 (which a user can't see publicly) and not in a module signature like Listing 8.

Here **sharing constraints** come into play as a solution.

Listing 10: `IntListStack` (Sharing Constraint Version)

```
1 module IntListStack : STACK with type element = int =  
2   struct  
3     exception EmptyStack  
4     type element = int  
5     type stack = int list  
6     let empty : stack = []  
7     (* ... remainder of the module ... *)  
8   end ;;
```

As shown in Listing 10, we add additional information to the type signature of module. `IntListStack` is a module of type `STACK` such that `type element = int`. Since, we've now specified the appropriate type of `type element` in the signature, the module can now be used.

Remark 4.2 (When to use Sharing Constraints). Sometimes students are confused about when it's appropriate to use sharing constraints.

We can think of abstract data types as data types that are hidden behind the abstraction barrier or "from the outside world". Users interact with these abstract data types by using functions and values defined by the module.

However, in some instances, we must or want to allow users to directly access abstract data types, such as with the `element` type for the `Stack` module signature.

Sharing constraints are intended to **share** information "hidden behind the abstraction barrier", so whenever we want to allow users to directly access or know more about an abstract data type, we should use a sharing constraint.

5 Functors

Definition 5.1 (Functor). The [OCaml documentation](#) defines a **functor** as a module that is parameterized by another module, similar to how functions are values parametrized by other values (i.e. arguments). In other words, we can think of a functor as a function that takes a module as input and produces a module as output.

Example 5.2 (Intervals). We will consider the `MakeInterval` example from lab. First, we define a module type called `ORDERED_TYPE`.

Listing 11: ORDERED_TYPE

```

1 module type ORDERED_TYPE =
2   sig
3     type t
4     val compare : t -> t -> int
5   end ;;
6
7   (* Modules satisfying the 'ORDERED_TYPE' signature will need to
8      define a type 't', as well as a function 'compare' that compares
9      two values for how they are ordered. *)
10
11 module IntOrderedType : ORDERED_TYPE =
12   struct
13     type t = int
14     let compare = Stdlib.compare
15   end ;;

```

Then, we define a module signature for intervals.

Listing 12: INTERVAL

```

1 module type INTERVAL =
2   sig
3     type interval
4     type endpoint
5     val create : endpoint -> endpoint -> interval
6     val is_empty : interval -> bool
7     val contains : interval -> endpoint -> bool
8     val intersect : interval -> interval -> interval
9   end ;;
10
11   (* Modules satisfying the 'INTERVAL' signature will allow working
12      with intervals of values of type 'endpoint'. *)

```

Functors can take modules satisfying one signature as input and generate modules satisfying a different signature as output. We will have our `MakeInterval` functor accept a module satisfying `ORDERED_TYPE` as input and produce a module satisfying `INTERVAL`.

Listing 13: MakeInterval

```

1 module MakeInterval (Endpoint : ORDERED_TYPE)
2   : (INTERVAL with type endpoint = Endpoint.t) =
3   struct
4     (* The functor defines a new module that contains a type 'interval',
5       which uses the 'Endpoint.t' type from the input module. *)
6     type interval =
7       | Interval of Endpoint.t * Endpoint.t
8       | Empty
9
10    let create (low : Endpoint.t)
11              (high : Endpoint.t)
12              : interval =
13      (* The functor also makes use of Endpoint.compare, a function
14        defined by the input module. *)
15      if Endpoint.compare low high > 0 then Empty
16      else Interval (low, high)
17      (* ... remainder of the module ... *)
18    end ;;
19
20    (* Here's a functor that takes a module with the 'ORDERED_TYPE'
21      signature and generates a module for an interval. *)
22    (* The input to the functor is a module 'Endpoint' that satisfies
23      the 'ORDERED_TYPE' signature. *)
24    (* The output of the functor is a module that satisfies the
25      'INTERVAL' signature. *)
26    (* We want the user to have access to the endpoint type, so we
27      add a sharing constraint. *)

```

Now, we can generate new modules by applying our functor to an argument module.

Listing 14: MakeInterval

```

1 module IntInterval =
2   MakeInterval (struct
3     type t = int
4     let compare = Stdlib.compare
5   end) ;;

```

Remark 5.3 (Functors with Many Parameters). Like functions, functors can have multiple arguments! We can even define higher-order functors that take in other functors, as you will see in Problem Set 6.

6 Practice Problems

Problem 6.1 (Inverting a Stack). Consider the polymorphic `Stack` module as shown in Listing 6.

1. Define a function `invert_stack : 'a Stack.stack -> 'a Stack.stack` that inverts the elements of the stack.
2. Write a few unit tests outside the `Stack` module testing the functionality of `invert_stack`.

Problem 6.2 (Graph). An (undirected) graph is a defined set of nodes and a set of edges, where each edge is a pair of different nodes.




Figure 1: Graph Example

There are many ways to represent a graph. For this problem, we will consider the **adjacency list** representation, in which we represent a finite graph as a collection of unordered lists such that each list represents the neighbors (vertices that share an edge) with a particular vertex. For instance, we can represent the graph in Figure 1 as: `[('g', ['h']); ('h', ['g']); ('b', ['c', 'f']); ('c', ['b', 'f']); ('f', ['b', 'c', 'k']); ('k', ['f']); ('d', [])]`.

Here, each element in the list is a pair, where the first element is a vertex and the second element is a list of neighbors.

We will work on defining a `Graph` module.

First, consider the following signatures for a `Vertex` and `Graph` module.

Listing 15: Vertex Module Signature

```
1 module type VERTEX =
2   sig
3     type t
4     val compare : t -> t -> int
5   end ;;
```

Listing 16: Graph Module Signature

```

1 module type GRAPH =
2 sig
3   type v
4   type graph
5   exception VertexAlreadyExists
6   exception VertexDoesNotExist
7   exception EdgeAlreadyExists
8
9   (* empty graph *)
10  val empty: graph
11
12  (* add a vertex to the graph *)
13  val addVertex : v -> graph -> graph
14
15  (* adds an edge between two vertices 'u' and 'v' *)
16  val addEdge : v -> v -> graph -> graph
17
18  (* Return a list of the vertices in the graph' *)
19  val vertices : graph -> v list
20
21  (* Return the neighbors of a vertex 'v' *)
22  val neighbors : v -> graph -> v list
23
24  (* Return the number of vertices and edges in the graph *)
25  val graph_size : graph -> int * int
26 end ;;

```

1. We will define a functor `MakeGraph` that takes in a module of type `Vertex` and returns a module of type `Graph`. What is the signature of this functor?
2. To define `MakeGraph`, start defining type `graph` based on the example given for Figure 1.
3. There are two main problems with this definition for `graph`. First, the `type` allows us to add the same vertex twice to the neighbors list for any vertex. Second, our list can contain the same vertex twice: for instance, the following is expressible `[('a', []); ('a', [])]`.

Let's address the first problem by representing the neighbors as a **set**, which is an unordered collection of unique elements. Use the OCaml [Set Module](#).

4. To address the second problem, suppose we instead represent the adjacency list using a **hash map** (or hash table). A hash map is a dictionary data structure in which we map unique keys to values. To do this, we'll use the OCaml [Map Module](#), taking the keys to be the vertex and the values to be the neighbors set, as defined in part 3.
5. Define `addVertex` and `addEdge`. If the user attempts to add a vertex or edge that already exists, return the appropriate exception.
6. Complete the implementation of the module by defining `vertices`, `neighbors`, and `graph_size`.

Problem 6.3 (Creating Graph Module).

1. Using the `MakeGraph` functor, define a module `IntGraph` such that the type of the vertices in the graph are integers.

Problem 6.4 (Set). Let's implement the `Set` module from scratch. Sets are an unordered list of elements with no duplicates.

1. Define a module signature for an `ORDERED_TYPE` which specifies the type of the elements, a function to compare elements, and a function to convert each of the elements into strings.
2. Create a module type for `Set` that includes the following values and operations on sets are supportable.
 - (a) `empty` : the empty set
 - (b) `add` : add an element to a set
 - (c) `take` : a function that takes an element in the set and returns the rest of the elements in the set as a pair : (h, t), where h is the extract element and t are the rest of the elements.
 - (d) `mem` : check if an element is a member of a set
 - (e) `union` : return the union of two sets
 - (f) `intersection` : return the intersection of two sets
 - (g) `print_set` : convert a set into a string.
3. Write a functor `MakeSet` that takes in a module of `ORDERED_TYPE` and returns a module of type `SET`.
4. Use `MakeSet` to define the following modules:
 - (a) A set of integers
 - (b) A set of strings
 - (c) A set of int lists
 - (d) A set of integer sets
5. Define a function `power_set` that returns a set of all the subsets of the original set. For this problem, you can create a function that returns the power set for a set of integers. For instance, the power set of $\{1, 2, 3, 4\}$ is

```
{ {}, {1}, {2}, {3}, {4}, {1, 2}, {1, 3}, {1, 4}, {2, 3}, {2, 4},  
  {3, 4}, {1, 2, 3}, {1, 3, 4}, {1, 2, 4}, {2, 3, 4}, {1, 2, 3,  
  4} }
```